

上限値/下限値を指定し成立範囲を解析区間として処理する

収録データ波形、又は演算データ波形から指定した上限値下限値範囲にある区間を検索処理する方法について説明します。

<対象チャネルを単数チャネルとして区間を検索する>

対象波形を指定した上限値下限値との比較演算により論理値化し、その論理波形を処理することで範囲を求めます。説明の都合上検索対象波形を収録チャネルの 2ch とします。

setep1: 対象波形を論理値化する

① 一定値を指定して論理値化する

説明の都合上、一定値を 10 として説明します。

\$1 “論理” = EQU(#2,10) /*\$1 は ch2 の値が 10 を”1”とした論理値*/

assign \$1 = 0,\$1,0 /*\$の先頭と終端に”0”を追加*/

② 上限値下限値を指定して論理値化する

説明の都合上、下限値を 10、上限値を 20 (範囲を 10<=範囲<=20) として説明します。

\$1 “論理” = AND(GTE(#2,10).LTE(#2,20)) /*\$1 は ch2 の値が 10~20 範囲が”1”とした論理値*/

assign \$1 = 0,\$1,0 /*\$の先頭と終端に”0”を追加

※ 求めた論理波形の先頭と終端に 0 を付加する理由は、論理波形の生成を比較演算によるレベル判定で行って求めている為、対象チャネルの先頭から 1 となった場合及び終端が 1 で終わった場合を考慮して強制的に遷移させることによります。その為、データ個数は+2 となっています。

※ 指定した区間が存在したか否かは SUM(\$1)> 1 で判定し、成立しない場合は以下の Step に進めません。

step2: 区間の開始データ番号、終了データ番号を求める

区間の開始データ番号/終了番号を求める方法は 2 通りあります。

① データ番号配列を生成し、step1 で求めた論理を差分演算した結果 (区間開始地点が”1”、区間終了地点+1 が”-1”となる波形) から、”1”の地点-1 のデータ番号を開始データ番号、”-1”の地点-2 を終了データ番号としてデータ番号配列から抽出方法

\$2 = LEN(#2)+2 /*\$2 は、ch2 のデータ個数+2*/

assign \$3 “index” = \$2<1> /*\$3 は、ch2 のデータ個数分の配列*/

\$3 = ACC(\$3)-1 /*\$3 は、0,1,2,3,4,5...のデータ番号配列*/

\$4 “区間開始 index” = ZSP(EQU(DIF(\$1),1),\$3)-1 /*\$4 は、区間開始データ番号*/

\$5 “区間終了 index” = ZSP(EQU(DIF(\$1),-1),\$3)-2 /*\$5 は、区間終了データ番号*/

② 論理遷移データ番号検索関数(LST 関数)を使用し論理波形から 0→1 への遷移地点-1 のデータ番号を開始データ番号、1→0 への遷移地点-2 のデータ番号を終了点データ番号として抽出する方法

\$4 “区間開始 index” = LST(1,\$1)-1 /*\$4 は、区間開始データ番号*/

\$5 “区間終了 index” = LST(0,\$1)-2 /*\$5 は、区間終了データ番号*/

step3: 抽出区間データ個数、区間数を求める

\$6 “区間データ個数” = \$5-\$4+1 /*\$6 は、区間のデータ個数*/

\$7 “区間数” = LEN(\$5) /*\$7 は、抽出区間数*/

\$8 “区間時間:sec” = (\$5-\$4)*PRDO /*\$8 は、区間時間*/

step4: 区間時間の長い順と発生時刻を求める

\$9 “区間時間の長い順” = SRT(2,\$8) /*\$9 は、区間時間の長い順に並べ替えた時間*/

\$10 “区間発生時刻:sec” = QUE(SRT(-2,\$8),\$4)*PRDO /*\$10 は、区間時間の長い順の発生時刻*/

step5: 区間の基本統計量を求める

\$11 “最大時間:sec” = MAX(\$8) /*\$11 は、区間の最も長い時間*/

\$12 “最小時間:sec” = MIN(\$8) /*\$12 は、区間の最も短い時間*/

\$13 “平均時間:sec” = MEA(\$8) /*\$13 は、区間の平均時間*/

\$14 “標準偏差値:sec” = STD(\$8) /*\$14 は、区間の標準偏差値*/

step6: 区間毎の ch2 の基本統計量を求める

read ch_name &1 &2 #2 ← /*ch2 の信号名と単位を取得*/

assign &3 = &1”_平均値”|&2 /*信号名と単位文字列生成*/

def ch_name \$14 &3 /*\$14 に名前と単位を付ける*/

assign \$14 = \$7<0> /*\$14 を区間数分の空配列生成*/

assign &3 = &1”_最大値”|&2 /*信号名と単位文字列生成*/

def ch_name \$15 &3 /*\$15 に名前と単位を付ける*/

assign \$15 = \$7<0> /*\$15 を区間数分の空配列生成*/

assign &3 = &1”_最小値”|&2 /*信号名と単位文字列生成*/

```

def ch_name $16 &3 /*$16 に名前と単位を付ける*/
assign $16 = $7<0> /*$16 を区間数分の空配列生成*/
assign &3 = &1"実効値"&2 /*信号名と単位文字列生成*/
def ch_name $17 &3 /*$17 に名前と単位を付ける*/
assign $17 = $7<0> /*$17 を区間数分の空配列生成*/
$18 "counter" = 0 /*格納先 index のカウンタ初期化*/
index $4 $6 /*演算範囲を指定する*/
proc statis
    $14($18) = MEA(#2) ← /*区間内の平均値を求める*/
    $15($18) = MAX(#2) ← /*区間内の最大値を求める*/
    $16($18) = MIN(#2) ← /*区間内の最小値を求める*/
    $17($18) = EFF(#2) ← /*区間内の実効値を求める*/
    $18 = $18+1 /*格納先 index のカウンタアップ*/
}statis
    
```

※ 他のチャンネルの統計量を求める場合は、記述中の#2 を書き換えることで良い。(上記 Script 文の“←”を付けた行)

step7: 指定したデータ番号範囲で指定したデータ個数以上の区間だけに整理する
 説明の都合上、指定するデータ番号の開始を 1000、終了を 5000、データ個数を 200 個以上として説明します。尚、指定可能なデータ番号は、開始データ番号<=MAX(\$4)、終了データ番号<=MAX(\$5)、及び、開始データ番号<=終了データ番号でなければなりません。

```
$19 = AND(AND(GTE($4,1000),LTE($5,5000)),GTE($6,200)) /*成立範囲を論理値"1"として求める*/
```

※ SUM(\$19)>0 でない場合は、指定した条件に指定した長さの区間が存在しません。存在しない場合は以下の演算は行えません。

```

$21 "成立区間の開始 index:" = ZSP($19,$4) /*$21 は、条件成立した区間開始データ番号*/
$22 "成立区間の終了 index:" = ZSP($19,$5) /*$22 は、条件成立した区間終了データ番号*/
$23 "成立区間データ個数:" = ZSP($19,$6) /*$23 は、条件成立した区間のデータ個数*/
    
```

<対象チャンネルを複数チャンネルとして区間を検索する>

対象チャンネルのそれぞれを指定した上限値下限値との比較演算により論理値化し、その論理波形を論理演算で統合した結果論理波形を処理することで範囲を求めます。

説明の都合上検索対象波形を収録チャンネルの 1ch、2ch、3ch とし、ch1 の上限値を 20、下限値を 10、ch2 の上限値を 100、下限値を 50、ch3 の上限値を 20、下限値を-20 とします。

step1: 対象波形を論理値化する

```

$1 "ch1 論理:" = AND(GTE(#1,10),LTE(#1,20))
$2 "ch2 論理:" = AND(GTE(#2,50),LTE(#2,100))
$3 "ch3 論理:" = AND(GTE(#3,-20),LTE(#3,20))
    
```


次に、論理値を統合する。

論理演算関数を使用して、論理値の統合を行う。

説明の都合上、ch1 の成立区間と ch2 の成立区間と ch3 の成立区間の全てが重複した区間(AND)とします。

```

$1 "論理:" = AND(AND($1,$2),$3)
assign $1 = 0,$1,0
    
```



以後の Step は、前述した Step と同じ処理となります。